

Principles of Linear Algebra With
Mathematica[®]
Linear Programming

Kenneth Shiskowski and Karl Frinkle

© Draft date March 12, 2012

Contents

1	Linear Programming	1
1.1	Geometric Linear Programming in Two Dimensions	1
1.2	Geometric Linear Programming in Three Dimensions	6
1.3	The Simplex Algorithm	12
1.4	Concluding Remarks	13

Chapter 1

Linear Programming

1.1 Geometric Linear Programming in Two Dimensions

This section is concerned with linear programming problems in two dimensions meaning the problem will have two variables called x and y so that we can plot in the xy -plane. A general linear programming problem in the two variables x and y involves maximizing or minimizing an *objective function*

$$z = \alpha x + \beta y + \delta \tag{1.1}$$

subject to *inequality constraints*

$$\begin{aligned} a_1x + b_1y &\leq c_1 \\ a_2x + b_2y &\leq c_2 \\ &\vdots \\ a_kx + b_ky &\leq c_k \\ &0 \leq x \\ &0 \leq y \end{aligned} \tag{1.2}$$

This type of problem occurs very often in industrial applications such as scheduling the optimal delivery of goods as well as the optimal production of goods. We will look at some practical examples after we discuss how such problems can be solved geometrically. Note that all but the last two inequalities are written as less than or equal inequalities since the direction of an inequality can always be changed by multiplication of the inequality by -1 . The easiest way to see which side of a line satisfies the inequality is to test the origin $(0, 0)$ in it and see if it is true or false.

We will begin by plotting in the xy -plane the feasible region of the problem which is the region simultaneously satisfying all of the inequality constraints. Each inequality constraint $ax + by \leq c$ generates a half-plane which is the side of the line $ax + by = c$ satisfying the inequality with the line itself. The feasible region will always lie in the first quadrant since $0 \leq x$ and $0 \leq y$ are always two of the constraints. The feasible region will be bounded by line segments or rays and so it is a polygonal region although not necessarily of finite area, and the boundary is part of the region.

Example 1.1.1. As an example, let's plot the region satisfying the system of inequalities

$$\begin{aligned} 3x - 5y &< 15, \\ -2x - 3y &< -18, \\ y &< 7, \\ 0 &\leq x, \\ 0 &\leq y. \end{aligned}$$

We shall plot the three lines $3x - 5y = 15$, $2x + 3y = 18$ and $y = 7$ only in the first quadrant attaching arrows to them to indicate the correct side of each which satisfies the inequality.

```
LinePlots = Plot[{{ $\frac{3}{5}x - 3$ ,  $-\frac{2}{3}x + 6$ , 7}, {x, 0, 20}, PlotStyle→{{
Thickness[0.007], Red}, {Thickness[0.007], Yellow},
{Thickness[0.007], Blue}}]
```




Figure 1.1: The three lines define a bounded region in the first quadrant.

```
ArrowPlots = Graphics[{Arrowheads[.05], Thickness[.010], Black,
Arrow[{{10, 3}, {8.97, 4.72}}], Black, Arrow[{{4, 10/3}, {5.11, 5.00}]}
```

```

}], Black, Arrow[{{7, 7}, {7, 5}}]];
Show[LinePlots, ArrowPlots]

```




Figure 1.2: Arrows pointing towards the region that satisfies the inequalities.

The feasible set R for this system of inequalities is the region of the first quadrant where all 3 arrows point simultaneously. The region R 's corner points or vertices are the 4 pairwise intersection points of these 4 line edges to the region. Let's find these 4 corner points and plot the polygon with them as vertices.

```
Soln1 = Solve[{3 x - 5 y == 15, 2 x + 3 y == 18}, {x, y}]
```

```
{ {x -> 135/19, y -> 24/19} }
```

```
Point1 = {x, y} /. Flatten[Soln1]
```

```
{ 135/19, 24/19 }
```

```
Soln2 = Solve[{3 x - 5 y == 15, y == 7}, {x, y}]
```

```
{ {x -> 50/3, y -> 7} }
```

```
Point2 = {x, y} /. Flatten[Soln2]
```

```
{ 50/3, 7 }
```

```
Point3 = {0, 7}; Point4 = {0, 6};
```

```
Region = Graphics[{Cyan, Polygon[{Point1, Point2, Point3, Point4}]}];
```

```
CornerPlot = ListPlot[{Point1, Point2, Point3, Point4}, PlotStyle→
Directive[Black, PointSize[Large]]];
```

```
TxtPtPlot = Graphics[{Text["Pt 1", {7.2, 0.55}], Text["Pt 2", {16.8,
6.2}], Text["Pt 3", {-0.75, 7.1}], Text["Pt 4", {-0.75, 5.9}]}];
```

```
Show[Region, LinePlots, ArrowPlots, CornerPlot, TxtPtPlot, Axes
→True, PlotRange→{{-3, 20}, {-3, 10}}]
```




Figure 1.3: Shaded region (cyan) satisfying all of the inequalities.

Now we want to solve the linear programming problem maximize $z = x + 2y + 5$ subject to the constraints above (which corresponds to the feasible region in Figure 1.3). We will plot some level curves of this objective function z , that is, we will plot $z = d$ for different constants d , in order to see where z achieves its maximum value in the region R above.

$$F[d_-, x_-] = \frac{d - x - 5}{2};$$

```
LevelCurves = Plot[Table[F[dval, x], {dval, 5, 40, 5}], {x, -3, 20},
PlotStyle→{{Thickness[0.004], Black}}];
```

```
Show[Region, LinePlots, ArrowPlots, CornerPlot, TxtPtPlot, Lev-
elCurves, Axes→True, PlotRange→{{-3, 20}, {-3, 10}}]
```




Figure 1.4: Feasible region and level curves (lines) of z .

It is clear from the above plot that the objective function z achieves its maximum at the right most corner point **point2** at $(\frac{50}{3}, 7)$. It also achieves its minimum value at the bottom corner point **point1** at $(\frac{135}{19}, \frac{24}{19})$. This situation is typical for all linear programming problems in that the points at which the objective function z achieves its maximum and minimum values over the feasible region R are corner points of the region. Let us now plug all four corner points into the objective function z to see that these two corner points give the largest and smallest values of z over all four corner points.

$$z = x + 2y + 5;$$

$$\mathbf{N}[z /. \mathbf{Soln1}][[1]]$$

14.6316

$$\mathbf{N}[z /. \mathbf{Soln2}][[1]]$$

35.6667

$$\mathbf{N}[z /. \{x \rightarrow \mathbf{Point3}[[1]], y \rightarrow \mathbf{Point3}[[2]]\}]$$

19.

$$\mathbf{N}[z /. \{x \rightarrow \mathbf{Point4}[[1]], y \rightarrow \mathbf{Point4}[[2]]\}]$$

17.

Mathematica has a command called **Maximize** which can be used to solve linear programming problems. The **Maximize** command makes partial use of the *simplex algorithm*, which is similar to **RowReduce** in that it uses pivoting on a matrix called a tableau to locate the corner points giving the objective function its largest or smallest values.

$$\mathbf{Maximize}[\{z, 3x - 5y \leq 15, -2x - 3y \leq -18, y \leq 7, x \geq 0, y \geq 0\}, \{x, y\}]$$

$$\left\{ \frac{107}{3}, \left\{ x \rightarrow \frac{50}{3}, y \rightarrow 7 \right\} \right\}$$

$$\mathbf{Minimize}[\{z, 3x - 5y \leq 15, -2x - 3y \leq -18, y \leq 7, x \geq 0, y \geq 0\}, \{x, y\}]$$

$$\left\{ \frac{278}{19}, \left\{ x \rightarrow \frac{135}{19}, y \rightarrow \frac{24}{19} \right\} \right\}$$

1.2 Geometric Linear Programming in Three Dimensions

This section is concerned with generalizing the previous section to linear programming problems in three dimensions meaning the problem will have three variables called x , y and z so that we can plot in space. A general linear programming problem in the three variables x , y , and z involves maximizing or minimizing an *objective function*

$$w = \alpha x + \beta y + \delta z + \theta \quad (1.3)$$

subject to *inequality constraints*

$$\begin{aligned} a_1x + b_1y + c_1z &\leq d_1 \\ a_2x + b_2y + c_2z &\leq d_2 \\ &\vdots \\ a_kx + b_ky + c_kz &\leq d_k \\ &0 \leq x \\ &0 \leq y \\ &0 \leq z \end{aligned} \quad (1.4)$$

We again wish to work through an example so that you can graphically see the feasible region R in the first octant which satisfies all of the inequality constraints. It will again be polygonal with the boundary of the region R consisting of plane sections where each corner point or vertex of the region R is a meeting point of three of these planes. Each inequality in this set of constraints determines a half-space where the inequality is true. The easiest way to see which side of a plane satisfies the inequality is to again test the origin $(0, 0, 0)$ in it and see if it is true or false.

Example 1.2.1. Let our feasible region R simultaneously satisfy the inequalities

$$\begin{aligned} 5x + 3y - 2z &\leq 30, \\ -7x - 4y - 3z &\leq -100, \\ z &\leq 25, \\ 0 &\leq x, \\ 0 &\leq y, \\ 0 &\leq z. \end{aligned}$$

We will attach arrows to each of these three planes which point into the feasible region R and also find the coordinates of the region's vertices.

```

Clear[z]

PlanarPlots = Plot3D[{{ $\frac{-30 + 5x + 3y}{2}$ ,  $\frac{100 - 7x - 4y}{3}$ , 25}},
{x, 0, 30}, {y, 0, 30}, PlotStyle→{{Thickness[0.007], Red},
{Thickness[0.007], Yellow}, {Thickness[0.007], Blue}}, Mesh→None,
AspectRatio→1, Lighting→"Neutral"];

v1 = {4, 15, 35/2}; d1 = {5, 3, -2};
v2 = {5, 5, 15}; d2 = {-3, -7, -4};
v3 = {6, 10, 25}; d3 = {0, 0, 4};

ArrowPlots = Graphics3D[{Arrowheads[.03], Thickness[.005], Black,
Arrow[{v1, v1 - d1/1.75}], Arrowheads[.03], Thickness[.005], Black,
Arrow[{v2, v2 - d2/1.75}], Arrowheads[.03], Thickness[.005], Black,
Arrow[{v3, v3 - d3}]}];

Show[ArrowPlots, PlanarPlots, Axes→True, PlotRange→{{0, 20},
{0, 30}, {0, 30}}, Lighting→"Neutral"]

```




Figure 1.5: Three planes and the directions which satisfy the inequalities.

Now we want to compute the corner points of this feasible region R . Each corner point is the intersection of three of the boundary planes of our region. We have six planes and must choose three of them at a time giving sixteen possibilities since we can not choose both $z = 25$ and $z = 0$ together. Now from the plot of the feasible region we see that there are only six corner points, not sixteen. Most of these triple linear systems of planes have no solution where

all variables are nonnegative.

Binomial[6, 3]

20

Planes = $\{-5x - 3y + 2z == -30, -7x - 4y - 3z == -100, z == 25, x == 0, y == 0, z == 0\}$;

Soln1 = **Solve**[{**Planes**[[1]], **Planes**[[2]], **Planes**[[5]]}, {**x**, **y**, **z**}

{**x** → 10, **y** → 0, **z** → 10}

Point1 = ({**x**, **y**, **z**} /. **Soln1**)[[1]]

{10, 0, 10}

Soln2 = **Solve**[{**Planes**[[1]], **Planes**[[2]], **Planes**[[4]]}, {**x**, **y**, **z**}

{**x** → 0, **y** → $\frac{290}{17}$, **z** → $\frac{180}{17}$ }

Point2 = ({**x**, **y**, **z**} /. **Soln2**)[[1]]

{0, $\frac{290}{17}$, $\frac{180}{17}$ }

Soln3 = **Solve**[{**Planes**[[1]], **Planes**[[3]], **Planes**[[4]]}, {**x**, **y**, **z**}

{**x** → 0, **y** → $\frac{80}{3}$, **z** → 25}

Point3 = ({**x**, **y**, **z**} /. **Soln3**)[[1]]

{0, $\frac{80}{3}$, 25}

Soln4 = **Solve**[{**Planes**[[1]], **Planes**[[3]], **Planes**[[5]]}, {**x**, **y**, **z**}

{**x** → 16, **y** → 0, **z** → 25}

Point4 = ({**x**, **y**, **z**} /. **Soln4**)[[1]]

{16, 0, 25}

Soln5 = **Solve**[{**Planes**[[2]], **Planes**[[3]], **Planes**[[4]]}, {**x**, **y**, **z**}

{**x** → 0, **y** → $\frac{25}{4}$, **z** → 25}

```
Point5 = ({x, y, z} /. Soln5)[[1]]
```

$$\left\{0, \frac{25}{4}, 25\right\}$$

```
Soln6 = Solve[{Planes[[2]], Planes[[3]], Planes[[5]]}, {x, y, z}]
```

$$\left\{\left\{x \rightarrow \frac{25}{7}, y \rightarrow 0, z \rightarrow 25\right\}\right\}$$

```
Point6 = ({x, y, z} /. Soln6)[[1]]
```

$$\left\{\frac{25}{7}, 0, 25\right\}$$

```
TxtPtPlot = Graphics3D[{Text["Pt 1", Point1], Text["Pt 2", Point2],
Text["Pt 3", Point3], Text["Pt 4", Point4], Text["Pt 5", Point5],
Text["Pt 6", Point6]}];
```

```
Show[ArrowPlots, PlanarPlots, TxtPtPlot, Axes→True, PlotRange
→{{0, 20}, {0, 30}, {0, 30}}, Lighting→"Neutral"]
```




Figure 1.6: Six corner points to the feasible region.

Now we have the six corner points or vertices of the feasible region R . Let's maximize and minimize the objective function $G = 9x + 2y + 5z + 13$. The level surfaces (parallel planes) of this objective function are $G = d$ for constants d . Let's plot five level surfaces of G . They again should indicate that the objective function w achieves its maximum and minimum values over the feasible

region at one of the region's vertices. The value of G or d increases as the level surfaces rise.

```
G[x_, y_, z_] = 9 x + 2 y + 5 z + 13;
LevelCurves = ContourPlot3D[Evaluate[Table[G[x, y, z] == d, {d,
50, 250, 50}]], {x, 0, 20}, {y, 0, 20}, {z, 0, 30}, Mesh→None, Contour-
Style→{{Opacity[0.5], Gray}}, AspectRatio→1/2];
Show[ArrowPlots, PlanarPlots, LevelCurves, Axes→True, Plot-
Range→{{0, 20}, {0, 30}, {0, 30}}, Lighting→"Neutral"]
```




Figure 1.7: Feasible region and level surfaces (planes) of w .

Before we optimize our objective function, we will graph the feasible region. To do this, one should first plot the points comprising the corners of each face of the boundaries of the feasible region. We will not display a graph of just the points, however the graph is useful in defining each side of the region R . Without this plot, the definitions of each side of the region (all six given next) may be confusing.

```
Region1 = Graphics3D[{Cyan, Polygon[{Point1,Point4,Point6}]}];
Region2 = Graphics3D[{Cyan, Polygon[{Point3,Point5,Point2}]}];
Region3 = Graphics3D[{Cyan, Polygon[{Point5, Point6, Point1,
Point2}]}];
Region4 = Graphics3D[{Cyan, Polygon[{Point3, Point4, Point6,
Point5}]}];
```

```

Region5 = Graphics3D[{Cyan, Polygon[{Point1, Point2, Point3,
Point4}]}];
Show[PlanarPlots, TxtPtPlot, Region1, Region2, Region3, Region4,
Region5, PlotRange→{{0, 20}, {0, 30}, {0, 30}}, Lighting→
"Neutral"]

```




Figure 1.8: Feasible region together with the planes.

```

N[Flatten[{G[x, y, z] /. Soln1, G[x, y, z] /. Soln2, G[x, y, z] /. Soln3,
G[x, y, z] /. Soln4, G[x, y, z] /. Soln5, G[x, y, z] /. Soln6}]]

```

```
{153., 100.059, 191.333, 282., 150.5, 170.143}
```

So the vertex which maximizes the objective function G is **point4** while the one which minimizes w is **point2**. Let's check this with the simplex algorithm. The simplex algorithm agrees with us that these points maximize and minimize G .

```

Maximize[{G[x, y, z], 5 x + 3 y - 2 z ≤ 30, -7 x - 4 y - 3 z ≤ -100, z ≤
25, x ≥ 0, y ≥ 0, z ≥ 0}, {x, y, z}]

```

```
{282, {x → 16, y → 0, z → 25}}
```

```

Minimize[{G[x, y, z], 5 x + 3 y - 2 z ≤ 30, -7 x - 4 y - 3 z ≤ -100, z ≤
25, x ≥ 0, y ≥ 0, z ≥ 0}, {x, y, z}]

```

```
{1701/17, {x → 0, y → 290/17, z → 180/17}}
```

1.3 The Simplex Algorithm

The simplex algorithm can be applied quite efficiently to even very large linear programming problems. It uses a matrix called a tableau which is similar to our augmented matrix used to solve a linear system and applies pivoting operations similar to what is used in **RowReduce**. It efficiently moves between the vertices of the feasible region looking for one which will maximize or minimize the objective function. We will not go into the details of the simplex algorithm any further as there are many excellent texts which discuss it in great detail if you need further information. The simplex algorithm is not difficult to implement but it is time consuming to explain why it works in full detail.

We will however do an example of using the simplex algorithm to solve the following story problem. A contractor builds ranch, colonial, two story and bi-level homes using plumbing, carpentry, electrical and masonry work. It takes 65 man-hours of plumbing, 185 man-hours of carpentry, 84 man-hours of electrical and 155 man-hours of masonry on average to complete one ranch home. It takes 95 man-hours of plumbing, 235 man-hours of carpentry, 175 man-hours of electrical and 70 man-hours of masonry on average to complete one colonial home. It takes 80 man-hours of plumbing, 255 man-hours of carpentry, 145 man-hours of electrical and 75 man-hours of masonry on average to complete one two story home. It takes 92 man-hours of plumbing, 215 man-hours of carpentry, 95 man-hours of electrical and 106 man-hours of masonry on average to complete one bi-level home. The contractor has available at most 2,500 man-hours of plumbing, 6,400 man-hours of carpentry, 3,250 man-hours of electrical and 3,600 man-hours of masonry work for a construction season. How many homes of each type should the contractor try to build in a season in order to maximize his season's total profit if he makes a profit on average of \$7,250 per ranch, \$8,875 per colonial, \$9,250 per two story and \$7,950 per bi-level?

This is a four variable linear programming problem with variables r = number of ranch homes to build, c = number of colonial homes to build, t = number of two story homes to build and b = number of bi-level homes to build. Then we want to maximize the total profit objective function

$$P = 7250r + 8875c + 9250t + 7950b$$

This is subject to the inequality constraints for plumbing:

$$65r + 95c + 80t + 92b \leq 2500,$$

for carpentry:

$$185r + 235c + 255t + 215b \leq 6400$$

for electrical:

$$84r + 175c + 145t + 95b \leq 3250$$

and for masonry:

$$155r + 70c + 75t + 106b \leq 3600$$

As well, $r \geq 0$, $c \geq 0$, $t \geq 0$ and $b \geq 0$.

$P = 7250 r + 8875 c + 9250 t + 7950 b$;

$N[\text{Maximize}[\{P, 65 r + 95 c + 80 t + 92 b \leq 2500, 185 r + 235 c + 255 t + 215 b \leq 6400, 84 r + 175 c + 145 t + 95 b \leq 3250, 155 r + 70 c + 75 t + 106 b \leq 3600, r \geq 0, c \geq 0, t \geq 0, b \geq 0\}, \{r, c, t, b\}]$

$\{242059., \{r \rightarrow 11.7063, c \rightarrow 4.84443, t \rightarrow 1.57593, b \rightarrow 12.5304\}\}$

This tells the contractor that they should and can build 11 ranch, 1 two story, 12 bi-level and 4 colonials in order to maximize total profit. Of course, some of these might be adjusted up by 1 (since we rounded each down) and you might also satisfy all of the constraints and get an even larger total profit.

1.4 Concluding Remarks

One of the most amazing things about using *Mathematica* to solve a problem like this is that you can also play with all of these numbers and see how it effects the answer. We also should point out that *Mathematica* has a **LinearProgramming** command, which allows you to write your linear optimization problem in terms of vector multiplications. Due to the complicated syntax involved in this command, mostly in regards to how \geq , $=$ and \leq are dealt with, we will not describe how to use the command here, however you may wish to explore this on your own. As a quick example, we can solve Example 1.2.1 with the following command:

$\text{LinearProgramming}[\{9, 2, 5\}, \{\{5, 3, -2\}, \{-7, -4, -3\}\}, \{\{30, -1\}, \{-100, -1\}\}, \{\{0, \text{Infinity}\}, \{0, \text{Infinity}\}, \{0, 25\}\}]$

$\{0, \frac{290}{17}, \frac{180}{17}\}$